Dérivation implicite

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. Le but de ce document est de transposer en Maple la dérivation implicite. On détaillera des résolutions habituelles de tangentes et de normales à des courbes définies implicitement. Dans ce document, lorsqu'il s'agit d'une équation de deux variables, nous supposerons toujours l'existence d'au moins une fonction implicite.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,implicitplot,pointplot,display,setoptions):
  #with(plottools,pointplot)
  setoptions(size=[400,400],scaling=constrained,axesfont=[times,roman,
  8],color=navy):
```

Équations implicites

Il est possible de dériver **implicitement** avec la macro-commande `diff`. À l'aide de la syntaxe fonctionnelle, il faut désigner dans l'équation, laquelle des deux variables sera désignée variable indépendante. Par exemple, dans le cas de l'équation ci-dessous,

$$y^2 + 5x = 3 - 5y^3$$

– soit $3y(x)^2 + 5x = 3 - 5y(x)^3$ si x est désignée variable indépendante,

– soit $3y^2 + 5x(y) = 3 - 5y^3$ si y est désignée variable indépendante.

Calculons $\frac{dy}{dx}$ où $y^2 + 5x = 3 - 5y^3$

```
> Équation:=3*y(x)^2+5*x=3-5*y(x)^3;
  diff(Équation,x);
```

$$\acute{E}quation := 3y(x)^2 + 5x = 3 - 5y(x)^3$$

$$6y(x)\left(\frac{d}{dx}y(x)\right) + 5 = -15y(x)^2\left(\frac{d}{dx}y(x)\right) \tag{2.1}$$

Reste maintenant à isoler y', c'est-à-dire $\frac{d}{dx}y(x)$.

```
>  Diff(y(x),x)=solve((2.1),diff(y(x),x));
```

$$\frac{d}{dx}y(x) = -\frac{5}{3y(x)(5y(x)+2)} \tag{2.2}$$

La macro-commande `implicitdiff` s'avère beaucoup plus souple dans le calcul des dérivées implicites et c'est la macro-commande qui sera retenue pour dériver implicitement dans la suite de ce document.

Macro-commande implicitdiff

La syntaxe de base est la suivante « `implicitdiff(Équation, var.dépendante, var.indépendante)` ».

Calculons de nouveau $\frac{dy}{dx}$ avec `implicitdiff`. sans utiliser la syntaxe fonctionnelle pour la variable dépendante, assignez $3y^2+5x=3-5y^3$ à `Équation` et calculons de nouveau $\frac{dy}{dx}$.

```
> Équation:=3*y^2+5*x=3-5*y^3;
  `y'`:=implicitdiff(Équation,y,x);
```

$$\text{Équation} := 3y^2+5x=-5y^3+3$$

$$y' := -\frac{5}{3y(5y+2)} \tag{3.1}$$

Il ne faut donc pas utiliser la notation fonctionnelle pour la variable y, celle qui est décrite implicitement par rapport à la variable x. On a donc que la macro-commande `implicitdiff` formule explicitement y', ce qui permet facilement d'assigner y' à un nom pour une éventuelle évaluation.

Évaluons, par exemple, $y' \left.\frac{dy}{dx}\right|_{\substack{x=55\\ y=-4}}$.

```
> Eval('`y'`',{x=55,y=-4})=eval(`y'`,{x=55,y=-4});
```

$$y'\Big|_{\{x=55,\,y=-4\}} = -\frac{5}{216} \tag{3.2}$$

Pour obtenir une dérivée implicite successive d'ordre supérieur, on procède de la même façon qu'avec la macro-commande `diff`. Soit, par exemple , calculons $\frac{d^3y}{dx^3}$ de l'exemple précédent $3y^2+5x=3-5y^3$. On l'obtient directement en utilisant l'opérateur de séquence `$`.

```
> Équation:=3*y^2+5*x=3-5*y^3;
  `y'''`:=implicitdiff(Équation,y,x$3);
```

$$\text{Équation} := 3y^2+5x=-5y^3+3$$

$$y''' := -\frac{250\,(125y^2+50y+6)}{27\,(3125y^5+6250y^4+5000y^3+2000y^2+400y+32)\,y^5} \tag{3.3}$$

Exemple 1

Calculons $\frac{dy}{dy}$ si $e^{xy}=x^2y^3$.

```
> Équation:=exp(x*y)=x^2*y^3;
```

```
`y'`:=implicitdiff(Équation,y,x);
```

$$\textit{Équation} := e^{xy} = x^2 y^3$$

$$y' := -\frac{y\left(-2xy^2 + e^{xy}\right)}{x\left(-3xy^2 + e^{xy}\right)} \quad \textbf{(3.1.1)}$$

Sur la base de l'égalité $e^{xy} = x^2 y^3$, exprimons la dérivée y', en termes de x et y seulement. Simplifions donc y' en substituant l'expression e^{xy} par $x^2 y^3$ dans la dérivée y'.

```
> ``=subs(Équation,(3.1.1));
```

$$= -\frac{y\left(x^2 y^3 - 2xy^2\right)}{x\left(x^2 y^3 - 3xy^2\right)} \quad \textbf{(3.1.2)}$$

Il est possible de transformer encore y' en factorisant xy au numérateur et au dénominateur. Pour terminer la simplification donc, normalisons la réponse précédente.

```
> normal((3.1.2));
```

$$= -\frac{y\left(xy - 2\right)}{x\left(xy - 3\right)} \quad \textbf{(3.1.3)}$$

Exemple 2

Voici un exemple classique de résolution d'un problème de tangentes.

Soit l'ellipse d'équation $4x^2 + 9y^2 - 8x + 54y + 49 = 0$. Calculer la pente de chacune des tangentes à cette courbe en $x = 3$ et obtener leur équation respective.

Dans un même graphique, tracer ces tangentes et la courbe. Enfin, déterminer, si possible, le point d'intersection de ces tangentes.

Solution proposée

Obtenons d'abord la forme canonique de l'équation de l'ellipse afin de bien mettre en évidence ses caractéristiques. C'est un travail nécessaire pour une approche judicieuse dans le tracé à faire.

À l'aide de la macro-commande `CompleteSquare` de la sous bibliothèque `withStudent[Precalculus]`, rendons accessible cette macro-commande pour la session Maple actuelle

```
> with(Student[Precalculus],CompleteSquare);
```

$$[\textit{CompleteSquare}] \quad \textbf{(3.2.1)}$$

```
> Équation:=4*x^2 + 9*y^2 - 8*x + 54*y + 49 = 0;
```

$$\textit{Équation} := 4x^2 + 9y^2 - 8x + 54y + 49 = 0 \quad \textbf{(3.2.2)}$$

```
> Éq1:=CompleteSquare(Équation);
```

$$\textit{Éq1} := 9\left(y+3\right)^2 + 4\left(x-1\right)^2 - 36 = 0 \quad \textbf{(3.2.3)}$$

```
> Éq2:=Éq1+(36=36);
```

(3.2.4)

$$\acute{E}q2 := 9\,(y+3)^2 + 4\,(x-1)^2 = 36 \qquad \textbf{(3.2.4)}$$

```
> Éq_Ellipse:=1/36*Éq2;
```

$$\acute{E}q\_Ellipse := \frac{(y+3)^2}{4} + \frac{(x-1)^2}{9} = 1 \qquad \textbf{(3.2.5)}$$

L'ensemble des points du plan cartésien dont les coordonnées vérifient l'équation $4x^2 + 9y^2 - 8x + 54y + 49 = 0$ est donc une ellipse centrée au point $(1,-3)$ dont l'axe principal est parallèle à l'axe des x. La demi-longueur de l'axe principal vaut $\sqrt{9} = 3$ et la demi-longueur du second axe vaut $\sqrt{4} = 2$, le tracé complet de cette ellipse sera obtenu avec $x \in [-2,-4]$ et $y \in [-5,-1]$

Spécifions l'option `view=[-3..5,-6..1]` pour mieux visualiser l'ellipse avec les axes de coordonnées.

```
> Ellipse:=implicitplot(Éq_Ellipse,x=-2..4,y=-5..-1):
  display(Ellipse,view=[-3..5,-6..1]);
```



Pour calculer les pentes des tangentes, il faut obtenir d'abord $\frac{dy}{dx}$ et ensuite, faire l'évaluation de la dérivée avec les coordonnées de chaque point de tangence.

Obtenons d'abord la dérivée $\frac{dy}{dx}$.

```
> `y'`:=implicitdiff(Équation,y,x);
```

$$y' := -\frac{4\,(x-1)}{9\,(y+3)} \qquad \textbf{(3.2.6)}$$

Pour évaluer la dérivée avec les deux coordonnées de chaque point de tangence, il faut obtenir l'ordonnée de chaque point de l'ellipse ayant pour abscisse la valeur 3.

```
> Sol:=solve(Équation,y);
  Ordonnées:=eval(Sol,x=3);
```

$$Sol := -3 + \frac{2\sqrt{-x^2+2x+8}}{3},\ -3 - \frac{2\sqrt{-x^2+2x+8}}{3}$$

$$Ordonn\acute{e}es := -3 + \frac{2\sqrt{5}}{3},\ -3 - \frac{2\sqrt{5}}{3} \qquad \textbf{(3.2.7)}$$

L'équation point-pente de chaque tangente est de la forme

$$y = \frac{dy}{dx}\Big|_{\substack{x = x_0 \\ y = y_0}} (x - x_0) + y_0$$

Obtenons la pente de chaque tangente en évaluant la dérivée avec les coordonnées des points de tangence respectif.

```
> m[1]:=eval(`y'`,{x=3,y=Ordonnées[1]});
  m[2]:=eval(`y'`,{x=3,y=Ordonnées[2]});
```

$$m_1 := -\frac{4\sqrt{5}}{15}$$

$$m_2 := \frac{4\sqrt{5}}{15} \qquad \textbf{(3.2.8)}$$

La pente de la tangente passant par le point $\left(3, \frac{-9+2\sqrt{5}}{3}\right)$ est $m_1 = -\frac{4\sqrt{5}}{15}$.

La pente de la tangente passant par le point $\left(3, \frac{-9-2\sqrt{5}}{3}\right)$ est $m_2 = \frac{4\sqrt{5}}{15}$

Les équations de chaque tangente sont alors les suivantes.

```
> Éq_tangente[1]:=y=m[1]*(x-3)+Ordonnées[1];
  Éq_tangente[2]:=y=m[2]*(x-3)+Ordonnées[2];
```

$$\acute{E}q\_tangente_1 := y = -\frac{4\sqrt{5}\,(x-3)}{15} - 3 + \frac{2\sqrt{5}}{3}$$

$$\acute{E}q\_tangente_2 := y = \frac{4\sqrt{5}\,(x-3)}{15} - 3 - \frac{2\sqrt{5}}{3} \qquad \textbf{(3.2.9)}$$

Pour obtenir la forme $y = mx + b$, il suffit d'appliquer la macro-commande `expand`.

```
> ÉqBis_tangente[1]:=expand(Éq_tangente[1]);
  ÉqBis_tangente[2]:=expand(Éq_tangente[2]);
```

$$\acute{E}qBis\_tangente_1 := y = -\frac{4\sqrt{5}\,x}{15} + \frac{22\sqrt{5}}{15} - 3$$

$$\acute{E}qBis\_tangente_2 := y = \frac{4\sqrt{5}\,x}{15} - \frac{22\sqrt{5}}{15} - 3 \qquad \textbf{(3.2.10)}$$

Superposons maintenant le tracé de l'ellipse avec ceux des deux droites tangentes.

```
> D_tangente[1]:=plot([x,rhs(Éq_tangente[1]),x=2..4],color=navy):
  D_tangente[2]:=plot([x,rhs(Éq_tangente[2]),x=2..4],color=navy):
  display({Ellipse,D_tangente[1],D_tangente[2]},view=[-3..5,-6..1]);
```



Finalement, déterminons les coordonnées du point d'intersection des deux droites tangentes en résolvant le système suivant.

```
> solve({Éq_tangente[1],Éq_tangente[2]});
```

$$\left\{x=\frac{11}{2}, y=-3\right\} \tag{3.2.11}$$

Le point d'intersection des tangentes est $\left(\frac{11}{2},\text{-}3\right)$.

```
> D_tangente[1]:=plot([x,rhs(Éq_tangente[1]),x=2..11/2],color=navy):
  D_tangente[2]:=plot([x,rhs(Éq_tangente[2]),x=2..11/2],color=navy):
  Point_t:=pointplot([11/2,-3],symbol=solidcircle,symbolsize=15):
  display({Point_t,Ellipse,D_tangente[1],D_tangente[2]},view=[-3..6,
  -6..1]);
```



Trouvons maintenant le point d'intersection des deux normales en ces points de tangence.

La pente de la normale passant par le point $\left(3, \frac{-9+2\sqrt{5}}{3}\right)$ est $-\frac{1}{m_1}$.

La pente de la tangente passant par le point $\left(3, \frac{-9-2\sqrt{5}}{3}\right)$ est $-\frac{1}{m_2}$

Les équations de chaque tangente sont alors les suivantes.

```
> Éq_normale[1]:=y=-1/m[1]*(x-3)+Ordonnées[1];
  Éq_normale[2]:=y=-1/m[2]*(x-3)+Ordonnées[2];
```

$$Éq\_normale_1 := y = \frac{3\sqrt{5}\,(x-3)}{4} - 3 + \frac{2\sqrt{5}}{3}$$

$$Éq\_normale_2 := y = -\frac{3\sqrt{5}\,(x-3)}{4} - 3 - \frac{2\sqrt{5}}{3} \qquad \textbf{(3.2.12)}$$

```
> D_normale[1]:=plot([x,rhs(Éq_normale[1]),x=2.5..3.5],color=navy):
  D_normale[2]:=plot([x,rhs(Éq_normale[2]),x=2.5..3.5],color=navy):
  display({Ellipse,D_normale[1],D_normale[2]},view=[-3..5,-6..1]);
```




```
> solve({Éq_normale[1],Éq_normale[2]});
```

$$\left\{x = \frac{19}{9}, y = -3\right\} \qquad \textbf{(3.2.13)}$$

Le point d'intersection des normales est $\left(\frac{19}{9}, -3\right)$

```
> D_normale[1]:=plot([x,rhs(Éq_normale[1]),x=19/9..3.5],color=navy):
  D_normale[2]:=plot([x,rhs(Éq_normale[2]),x=19/9..3.5],color=navy):
  Point_n:=pointplot([19/9,-3],symbol=solidcircle,symbolsize=15):
  display({Point_n,Ellipse,D_normale[1],D_normale[2]},view=[-3..5,-6.
  .1]);
```




```
> display({Ellipse,
        Point_t,D_tangente[1],D_tangente[2],
        Point_n,D_normale[1],D_normale[2]},
        view=[-3..6,-6..1]);
```



Exemple 3

Soit l'équation $y^3 + y^2 - 5y - x^2 = -4$. Traçons le lieu définie par cette équation.

```
> Éq:=y^3+y^2-5*y-x^2=-4;
  Lieu:=implicitplot(Éq,x=-5..5,y=-5..5);
```

$$Éq := y^3 - x^2 + y^2 - 5y = -4$$



Trouvons les points de ce tracé où les tangentes sont verticales. Trouvons donc les points où la dérivée n'existe pas.

```
> `y'`:=implicitdiff(Éq,y,x);
```

$$y' := \frac{2x}{3y^2+2y-5} \qquad \textbf{(3.3.1)}$$

La dérivée n'existe pas lorsque le dénominateur s'annule.

```
> Sol:=solve(denom((3.3.1))=0,{y});
```

$$Sol := \{y=1\}, \left\{y=-\frac{5}{3}\right\} \qquad \textbf{(3.3.2)}$$

Obtenons les abcisses des points du tracé qui ont 1 et $-\frac{5}{3}$ comme ordonnées. Selon le graphique, nous devrions en trouver deux pour chaque valeur de y.
Substituons la valeur de $y=1$ et $y=-\frac{5}{3}$ dans $y^3+y^2-5y-x^2+4=0$ puis résolvons pour x chaque équation obtenue.

```
> solve(subs(y=1,y^3-x^2+y^2-5*y+4=0),{x});
  solve(subs(y=-5/3,y^3-x^2+y^2-5*y+4=0),{x});
```

$$\{x=-1\}, \{x=1\}$$

$$\left\{x=-\frac{\sqrt{849}}{9}\right\}, \left\{x=\frac{\sqrt{849}}{9}\right\} \qquad \textbf{(3.3.3)}$$

Les points recherchés sont donc $\left(-\frac{\sqrt{849}}{9},-\frac{5}{3}\right)$, $(-1,-1)$, $(1,1)$ et $\left(\frac{\sqrt{849}}{9},-\frac{5}{3}\right)$.

Traçons finalement les tangentes au grahique en ces points.

```
> Tangente_1:=plot([-sqrt(849)/9,t,t=-3..-0.5],color="Niagara 16",
  thickness=1):
  Tangente_2:=plot([-1,t,t=0..2],color="Niagara 16",thickness=1):
  Tangente_3:=plot([1,t,t=0..2],color="Niagara 16",thickness=1):
  Tangente_4:=plot([sqrt(849)/9,t,t=-3..-0.5],color="Niagara 16",
  thickness=1):
```

```
display(Lieu,Tangente_||(1..4));
```